
PINCH_RESISTOR
*
*                XRpinch
*     Vtest   ________
*       _____|_/\  /\  /\_
*     _|__       \/  \/  _|_
*    /_   \      :       ///
*   // \   \     :
*   \   \_//     :        Rset
*    \____/      :./\  /\  /\_
*      _|_           \/  \/  _|_
*      ///     VRset         ///
*    Vramp

Vramp    Vtest  0    PWL(   0     0.1  5  5.0)
XRpinch  Vtest  0    VRset  Rpinch
Rset     VRset  0    1k
.tran    100m   5     0     100m

.control
run
plot  -vramp#branch
dump
.endc

*   ___                ___   .SUBCKT  Rpinch   INP   INN  VRS
*  |INP|_______       |VRS|
*  |___|  |   _|_     |___|          ___   ________     ___
*         |_ /   \     _|_    ==>   |INP|_|_/\  /\  /\_|INN|
*           |Bpinc|   / _ \         |___|     \/  \/   |___|
*         __|     |   \/ \/                  _:_
*         |  \___/    /\_/\                 |VRS|
*         |    |      \___/Isen             |___|
*   ___   |    |       _|_
*  |INN|__|____|       ///
*  |___|
*

.SUBCKT  Rpinch   INP   INN  VRS
Isen     VRS      0     -1u
Bpinc    INP   INN   I = .000001*(v(INP) - v(INN))/(v(VRS)*(1+1*(v(INP) - v(INN))))
.ENDS    Rpinch
*
*      Attach Pin 2 to highest voltage
*
*         ________________________________________________________
*        /               _____________________      Nepi         /
*       /  ____________ /..................../_______________   /
*      / / ________    /                    /     ________  /  /
*     / / /____ 1 /   /         2          /     /____ 3 / /  /
*    / / //___/| /   /                    /     //___/| / /  /
*   / / / |___|//   /                    /     / |___|// /  /
* _/_/_/_______/___/____________________/_____/_______/_/__/___
*  | | | p+imp |  |       n+imp         |     | p+imp | |  |
*  |  \ \_____/    \___________________/ base  \_____/ /   |
*  |   \______________________________________________/    | Psub
*  |  _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _  _ _ _ _ _ _ _ |
*  |/   N BURIED LAYER                                    \|
*   \ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ /
*          P-Substrate
*

.end




